
    Date: Tuesday, January 5, 2021 @ 19:51:51
  Author: andyrtr
Revision: 405371

archrelease: copy trunk to testing-x86_64

Added:
  qpdf/repos/testing-x86_64/
  qpdf/repos/testing-x86_64/PKGBUILD
    (from rev 405370, qpdf/trunk/PKGBUILD)

----------+
 PKGBUILD |   49 +++++++++++++++++++++++++++++++++++++++++++++++++
 1 file changed, 49 insertions(+)

Copied: qpdf/repos/testing-x86_64/PKGBUILD (from rev 405370, qpdf/trunk/PKGBUILD)
===================================================================
--- testing-x86_64/PKGBUILD	                        (rev 0)
+++ testing-x86_64/PKGBUILD	2021-01-05 19:51:51 UTC (rev 405371)
@@ -0,0 +1,49 @@
+# Maintainer: AndyRTR <andyrtr at archlinux.org>
+# Contributor: Miguel Revilla <yo at miguelrevilla.com>
+# Contributor: David Sotelo <dvsotelo at gmail.com>
+# Contributor: Nuno Araujo <nuno.araujo at russo79.com>
+
+pkgname=qpdf
+pkgver=10.1.0
+pkgrel=1
+pkgdesc="QPDF: A Content-Preserving PDF Transformation System"
+arch=('x86_64')
+url="https://github.com/qpdf/qpdf"
+license=('custom:Artistic-2.0' 'Apache')
+depends=('libjpeg-turbo' 'zlib' 'gnutls')
+#optdepends=('perl:	required for fix-qdf')
+#checkdepends=('perl')
+# https://github.com/qpdf/qpdf/releases/download/release-qpdf-10.1.0/qpdf-10.1.0.sha256
+source=(#https://downloads.sourceforge.net/${pkgname}/${pkgname}-${pkgver}.tar.gz{,.asc})
+        https://github.com/qpdf/qpdf/releases/download/release-${pkgname}-${pkgver}/${pkgname}-${pkgver}.tar.gz{,.asc})
+sha256sums=('862c144e4516302327cea908f2879131cc8198b10d3d3a90ef7bc006a915120d'
+            'SKIP')
+validpgpkeys=('C2C96B10011FE009E6D1DF828A75D10998012C7E') # "Jay Berkenbilt <ejb at ql.org>"
+
+build() {
+  cd ${pkgname}-${pkgver}
+  ./configure --prefix=/usr \
+    --disable-static \
+    --enable-crypto-gnutls \
+    --disable-crypto-openssl \
+    --disable-implicit-crypto \
+    --enable-show-failed-test-output
+  sed -i -e 's/ -shared / -Wl,-O1,--as-needed\0/g' libtool
+  make
+}
+
+check() {
+  cd ${pkgname}-${pkgver}
+  make check # passes all
+}
+
+package() {
+  cd ${pkgname}-${pkgver}
+  make DESTDIR="${pkgdir}/" install
+
+  install -Dm644 completions/bash/qpdf "${pkgdir}/usr/share/bash-completion/completions/qpdf"
+  install -Dm644 completions/zsh/_qpdf "${pkgdir}/usr/share/zsh/site-functions/_qpdf"
+
+  mkdir -m755 -p "${pkgdir}"/usr/share/licenses/${pkgname}
+  install -m644 {Artistic-2.0,LICENSE.txt} "${pkgdir}"/usr/share/licenses/${pkgname}/
+}

